

Points: Geologic Foliation Dip and Strike

Query generates a foliation symbol based on dip, strike and dip direction. This query is data dependent, change *foliation.strike*, *foliation.dip* and *foliation.dip_direction* to values appropriate for your data.

```

# This query implements the geologic
# strike/dip for foliation includes special case
# of vertical foliation
LineStyleDropAnchor(0) # center of symbol
LineStyleSetColor(0,0,0) #black symbols
LineStyleSetCapJoinType(1,1) #squared off
strike1 = foliation.strike
dip1 = foliation.dip
scaling = 7
radius = 4*scaling
LineStyleSetLineWidth(scaling)
textheight = 6 *scaling
#Draw Strike Line
if(dip1!=0){
  LineStyleLineTo(strike1,16*scaling)
  LineStyleMoveToAnchor(0)
  LineStyleLineTo(strike1 - 180,16*scaling)
  LineStyleMoveToAnchor(0)
}
#rotate symbol by 180 when dip direction indicates
if (foliation.dip_direction) rot = 180 else rot = 0
strikeangle =strike1+rot
if (strikeangle > 360) {strikeangle = strikeangle -360 }
markangle = strikeangle -90
if (markangle < 0) {markangle = markangle + 360}
sidesize=8*scaling

#design the dip direction triangle
LineStyleMoveToAnchor(0)
LineStyleMoveTo (strike1 + rot - 180 ,sidesize/2) # starting position
LineStyleRecordPolygon(1)
LineStyleMoveTo(strikeangle-60,sidesize)
LineStyleMoveTo(strikeangle+60 ,sidesize)
if(dip1==90){ #extend triangle into a diamond for vertical foliation
  LineStyleMoveTo(strikeangle+120,sidesize)
  LineStyleMoveTo(strikeangle-120,sidesize)
}
else {
  LineStyleMoveTo(strikeangle-180,sidesize)
}
LineStyleRecordPolygon(0)
LineStyleDrawPolygon(1) #draw triangle or diamond

if (dip1<90) { #label non-vertical foliations
  LineStyleMoveToAnchor(0)
  LineStyleMoveTo(markangle,sidesize/1.4)
  LineStyleSetTextColor(0,0,0)
  str$ = sprintf("%2d",dip1);
  if ((markangle>90) and (markangle<270)){
    LineStyleTextNextPosition(str$,textheight,0,0,nextx,nexty,tlength)
    LineStyleMoveTo(180,tlength)
  }
}
if (markangle>180){
  LineStyleMoveTo(- 90,textheight)
  } LineStyleDrawText(str$,textheight,0,2)
}

```

